

PING PONG

JIM BUTTERFIELD

Play against the computer, or change the program for a two-player game. On each shot, you choose between four plays: Spin, Lob, Block, or Slam. If you're playing the left side of the court, use the left-hand buttons (0, 4, 8 and C). See the diagram at right.

C	slam	F
8	block	B
4	lob	7
0	spin	3

Each shot has its own strengths and weaknesses: for example, a Slam is a powerful shot, but it's also likely to be "fluffed". Strategy is not trivial - your chances of success on any play depend not only on your choice of shot, but on what shots have gone before. You'll have to learn the combinations the hard way.

You'll see the net in the middle of the court. Don't try to play the ball until it is on your side of the net, or you'll lose the point. Each type of shot has a distinctive appearance, which you'll learn to recognize. They are similar to the key positions: a Spin lights the bottom segment, a Lob lights the middle segment, a Block lights the upper segment, and the mighty Slam shot lights all three segments and travels faster.

The original version of the game was published for the HP-67 calculator in "65 Notes", V4N2P5. Authorship was not given.

At first, the shots will come too fast for you to cope with. There are two ways to solve this. The easy way is the "freeze" the ball by holding down any unused key, like AD or 7: play will be suspended until you figure out what you want to do next. The harder way, but not too hard, is just to slow down the ball by changing the program: locations 0331 to 0334 contain the speeds for each type of shot. Increase these values and the ball will slow down, e.g., 40 40 40 28 will halve the speed.

For a two-player game, where KIM does not play the right side, change location 032C to 01. To have KIM play the left side, change location 032B to 00. KIM plays a strong game, but CAN BE BEATEN!

```

0200 20 40 1F START JSR KEYIN    directional registrs
0203 20 6A 1F      JSR GETKEY    input key
0206 C9 13        CMP #$13      GO key?
0208 D0 0A        BNE NOGO      nope, skip
                                ; GO key - set up game here
020A A2 08        LDX #8        get 9 ..
020C BD 24 03 SETUP LDA INIT,X   ..initial valus
020F 95 80        STA SPEED,X    to zero page
0211 CA          DEX
0212 10 F8        BPL SETUP
                                ; test legal keys (0,3,4,7,8,B,C,F)
0214 C9 10        NOGO CMP #$10  key 0 to F?
0216 B0 22        BCS NOKEY     no, skip
0218 AA          TAX            save key in X
0219 29 03        AND #3       test column
021B F0 04        BEQ KEY      col 0 (0,4,8,C)?
021D C9 03        CMP #3       col 3 (3,7,B,F)?
021F D0 19        BNE NOKEY     neither - skip
0221 45 85        KEY  EOR PLACE check vs ball postn
0223 A8          TAY
0224 29 04        AND #4       ball off screen?
0226 D0 12        BNE NOKEY
0228 8A          TXA            restore key
0229 45 84        EOR DIRECT   ball going away?
022B 29 02        AND #2
022D F0 0B        BEQ NOKEY     yes, ignore key
022F 98          TYA            ball position
0230 29 02        AND #2       wrong side of net?
0232 D0 69        BNE POINT     yes, lose!
                                ; legal play found here
0234 8A          TXA            restore key
0235 4A 4A        LSRA LSRA     type (0=Spin, etc)
0237 20 B1 02      JSR SHOT     make shot
                                ; key rtns complete - play ball
023A 20 40 1F NOKEY JSR KEYIN   if key still prest..
023D D0 27        BNE FREEZE    freeze ball
023F C6 83        DEC PAUSE
0241 10 23        BPL FREEZE    wait til timeout
0243 A5 80        LDA SPEED
0245 85 83        STA PAUSE
0247 18          CLC
0248 A5 85        LDA PLACE     move..
024A 65 84        ADC DIRECT    ..ball
024C 85 85        STA PLACE
024E 29 04        AND #4       ball still..
0250 F0 14        BEQ FREEZE    in court?
                                ; ball outside - KIM to play?
0252 A5 85        LDA PLACE
0254 30 04        BMI TESTL     ball on left
0256 A5 88        LDA PRITE     KIM plays right?
0258 10 02        BPL SKPT      unconditional
025A A5 87        TESTL LDA PLEFT KIM plays left?
025C D0 3F        SKPT BNE POINT no, lose point

```

```

; KIM plays either side here
025E A6 82          LDX LOG      log determines..
0260 BD 39 03      LDA PLAY,X    ..KIM's play
0263 20 B1 02      JSR SHOT      make the shot
0266 A9 7F        FREEZE LDA #$7F
0268 8D 41 17      STA PADD      open registers
; light display here
026B A0 13        LDY #$13
026D A2 01        LDX #1
026F 86 89        STX DIGIT      count score digts
0271 A5 86        LDA SCORE
0273 4A 4A        LSRA LSRA      shift & store..
0275 4A 4A        LSRA LSRA      ..left player score
0277 85 8A        STA ARG
0279 A5 86        LDA SCORE
027B 29 0F        AND #$0F      ..right player score
027D AA          TAX
027E BD E7 1F HOOR LDA TABLE,X
0281 20 A4 02      JSR SHOW
0284 A6 8A        LDA ARG
0286 C6 89        DEC DIGIT
0288 10 F4        BPL HOOP
028A A2 03        LDX #3
028C BD 2D 03 VUE  LDA PIX,X
028F E4 85        CPX PLACE
0291 D0 02        BNE NOPIX
0293 05 81        ORA SPOT      show the ball
0295 20 A4 02 NOPIX JSR SHOW
0298 CA          DEX
0299 10 F1        BPL VUE
029B 30 03        BMI SLINK
; lose! score & reverse board
029D 20 E9 02 POINT JSR SKORE
02A0 D8          SLINK CLD
02A1 4C 00 02      JMP START    return to main loop
; display subroutine
02A4 8D 40 17 SHOW STA SAD
02A7 8C 42 17      STY SBD
02AA C6 8B        STALL DEC MOD
02AC D0 FC        BNE STALL
02AE 88 88        DEY DEY
02B0 60          RTS
02B1 A8          SHOT TAY      save shot in Y
02B2 A6 82        LDX LOG      old log in X
02B4 06 82        ASL LOG
02B6 06 82        ASL LOG
02B8 05 82        ORA LOG
02BA 29 0F        AND #$F      update log book
02BC 85 82        STA LOG      ..last two shots
02BE 38          SEC
02BF A5 80        LDA SPEED
02C1 E5 83        SBC PAUSE    invert timing
02C3 85 83        STA PAUSE

```

```

; set speed & display segment(s)
02C5 B9 31 03 LDA SPD,Y
02C8 85 80 STA SPEED
02CA B9 35 03 LDA SEG,Y
02CD 85 81 STA SPOT

; test play success - random
02CF BD 49 03 LDA CHANCE,X odds from log bk
02D2 88 GIT DEY
02D3 30 04 BMI GET
02D5 4A 4A LSRA LSRA
02D7 10 F9 BPL GIT unconditional
02D9 29 03 GET AND #3 odds 0 to 3..
02DB 0A ASL A now 0 to 6
02DC 85 8C STA TEMP
02DE AD 04 17 LDA TIMER random number
02E1 29 07 AND #7 now 0 to 7
02E3 C5 8C CMP TEMP
02E5 F0 33 BEQ REVRs success?
02E7 90 31 BCC REVRs success?

; lose a point & position to serve
02E9 A2 04 SKORE LDX #4 position ball R
02EB A5 84 LDA DIRECT
02ED 0A 0A ASLA ASLA
02EF 0A 0A ASLA ASLA
02F1 10 04 BPL OVER
02F3 A2 FF LDX #$FF position ball L
02F5 A9 01 LDA #1
02F7 86 85 OVER STX PLACE
02F9 18 CLC
02FA 65 86 ADC SCORE
02FC 85 86 STA SCORE
02FE A0 00 LDY #0 end game, kill ball
0300 AA TLP TAX
0301 29 0F AND #$F get one score
0303 C9 0B CMP #$11 11 points?
0305 D0 02 BNE SKI
0307 84 84 STY DIRECT kill ball
0309 8A SKI TXA
030A 4A 4A LSRA LSRA
030C 4A 4A LSRA LSRA
030E D0 F0 BNE TLP

; set serve - speed, spot, log, pause
0310 A2 03 LDX #3
0312 BD 24 03 SRV LDA INIT,X
0315 95 80 STA SPEED,X
0317 CA DEX
0318 10 F8 BPL SERVE

; reverse ball direction
031A A5 84 REVRs LDA DIRECT
031C 18 CLC
031D 49 FF EOR #$FF
031F 69 01 ADC #1
0321 85 84 STA DIRECT
0323 60 RTS

```

```

; tables - in Hexadecimal format
0324 INIT      30 08 00 80 01 FF 00 01 00
032D PIX       00 06 30 00
0331 SPD       20 20 20 14
0335 SEG       08 40 01 49
0339 PLAY      02 02 01 02 01 03 01 02 03 03 00 02 00 00 02 02
0349 CHANCE    78 B5 9E 76 6E A1 AE 75 AA EB 8F 75 5B 56 7A 35
0359 end

```

```

Zero Page:  80: SPEED - speed ball travels
            81: SPOT - segment(s) ball lights
            82: LOG - record of recent plays
            83: PAUSE - delay before ball moves
            84: DIRECT - direction of ball
            85: PLACE - position of ball
            86: SCORE
            87: PLEFT - 0 for KIM to play left
            88: PRITE - 0 for KIM to play right

```

***** Hex Dump - Ping Pong *****

```

0200 20 40 1F 20 6A 1F C9 13 D0 0A A2 08 BD 24 03 95
0210 80 CA 10 F8 C9 10 B0 22 AA 29 03 F0 04 C9 03 D0
0220 19 45 85 A8 29 04 D0 12 8A 45 84 29 02 F0 0B 98
0230 29 02 D0 69 8A 4A 20 B1 02 20 40 1F D0 27 C6
0240 83 10 23 A5 80 85 83 18 A5 85 65 84 85 85 29 04
0250 F0 14 A5 85 30 04 A5 88 10 02 A5 87 D0 3F A6 82
0260 BD 39 03 20 B1 02 A9 7F 8D 41 17 A0 13 A2 01 86
0270 89 A5 86 4A 4A 4A 85 8A A5 86 29 0F AA BD E7
0280 1F 20 A4 02 A6 8A C6 89 10 F4 A2 03 BD 2D 03 E4
0290 85 D0 02 05 81 20 A4 02 CA 10 F1 30 03 20 E9 02
02A0 D8 4C 00 02 8D 40 17 8C 42 17 C6 8B D0 FC 88 88
02B0 60 A8 A6 82 06 82 06 82 05 82 29 0F 85 82 38 A5
02C0 80 E5 83 85 83 B9 31 03 85 80 B9 35 03 85 81 BD
02D0 49 03 88 30 04 4A 4A 10 F9 29 03 0A 85 8C AD 04
02E0 17 29 07 C5 8C F0 33 90 31 A2 04 A5 84 0A 0A 0A
02F0 0A 10 04 A2 FF A9 01 86 85 18 65 86 85 86 A0 00

0300 AA 29 0F C9 0B D0 02 84 84 8A 4A 4A 4A 4A D0 F0
0310 A2 03 BD 24 03 95 80 CA 10 F8 A5 84 18 49 FF 69
0320 01 85 84 60 30 C0 00 80 01 FF 00 01 00 00 06 30
0330 00 20 20 20 14 08 40 01 49 02 02 01 02 01 03 01
0340 02 03 03 00 02 00 00 02 02 78 B5 9E 76 6E A1 AE
0350 75 AA EB 8F 75 5B 56 7A 35

```